<commit_before>{ config, lib, pkgs, ... }:

with lib;

let
  cfg = config.services.corerad;

  writeTOML = name: x:
    pkgs.runCommandNoCCLocal name {
      passAsFile = ["config"];
      config = builtins.toJSON x;
      buildInputs = [ pkgs.go-toml ];
    } "jsontoml < $configPath > $out";

in {
  meta.maintainers = with maintainers; [ mdlayher ];

  options.services.corerad = {
    enable = mkEnableOption "CoreRAD IPv6 NDP RA daemon";

    settings = mkOption {
      type = types.uniq types.attrs;
      example = literalExample ''
        {
          interfaces = [
            # eth0 is an upstream interface monitoring for IPv6 router advertisements.
            {
              name = "eth0";
              monitor = true;
            }
            # eth1 is a downstream interface advertising IPv6 prefixes for SLAAC.
            {
              name = "eth1";
              advertise = true;
              prefix = [{ prefix = "::/64"; }];
            }
          ];
          # Optionally enable Prometheus metrics.
          debug = {
            address = "localhost:9430";
            prometheus = true;
          };
        }
      '';
      description = ''
        Configuration for CoreRAD, see <link xlink:href="https://github.com/mdlayher/corerad/blob/master/internal/config/default.toml"/>
        for supported values. Ignored if configFile is set.
      '';
    };

    configFile = mkOption {
      type = types.path;
      example = literalExample "\"\${pkgs.corerad}/etc/corerad/corerad.toml\"";
      description = "Path to CoreRAD TOML configuration file.";
    };

    package = mkOption {
      default = pkgs.corerad;
      defaultText = literalExample "pkgs.corerad";
      type = types.package;
      description = "CoreRAD package to use.";
    };
  };

  config = mkIf cfg.enable {
    # Prefer the config file over settings if both are set.
    services.corerad.configFile = mkDefault (writeTOML "corerad.toml" cfg.settings);

    systemd.services.corerad = {
      description = "CoreRAD IPv6 NDP RA daemon";
      after = [ "network.target" ];
      wantedBy = [ "multi-user.target" ];
      serviceConfig = {
        LimitNPROC = 512;
        LimitNOFILE = 1048576;
        CapabilityBoundingSet = "CAP_NET_ADMIN CAP_NET_RAW";
        AmbientCapabilities = "CAP_NET_ADMIN CAP_NET_RAW";
        NoNewPrivileges = true;
        DynamicUser = true;
        ExecStart = "${getBin cfg.package}/bin/corerad -c=${cfg.configFile}";
        Restart = "on-failure";
      };
    };
  };
}<commit_msg>nixos/corerad: set systemd unit Type=notify<commit_after>{ config, lib, pkgs, ... }:

with lib;

let
  cfg = config.services.corerad;

  writeTOML = name: x:
    pkgs.runCommandNoCCLocal name {
      passAsFile = ["config"];
      config = builtins.toJSON x;
      buildInputs = [ pkgs.go-toml ];
    } "jsontoml < $configPath > $out";

in {
  meta.maintainers = with maintainers; [ mdlayher ];

  options.services.corerad = {
    enable = mkEnableOption "CoreRAD IPv6 NDP RA daemon";

    settings = mkOption {
      type = types.uniq types.attrs;
      example = literalExample ''
        {
          interfaces = [
            # eth0 is an upstream interface monitoring for IPv6 router advertisements.
            {
              name = "eth0";
              monitor = true;
            }
            # eth1 is a downstream interface advertising IPv6 prefixes for SLAAC.
            {
              name = "eth1";
              advertise = true;
              prefix = [{ prefix = "::/64"; }];
            }
          ];
          # Optionally enable Prometheus metrics.
          debug = {
            address = "localhost:9430";
            prometheus = true;
          };
        }
      '';
      description = ''
        Configuration for CoreRAD, see <link xlink:href="https://github.com/mdlayher/corerad/blob/master/internal/config/default.toml"/>
        for supported values. Ignored if configFile is set.
      '';
    };

    configFile = mkOption {
      type = types.path;
      example = literalExample "\"\${pkgs.corerad}/etc/corerad/corerad.toml\"";
      description = "Path to CoreRAD TOML configuration file.";
    };

    package = mkOption {
      default = pkgs.corerad;
      defaultText = literalExample "pkgs.corerad";
      type = types.package;
      description = "CoreRAD package to use.";
    };
  };

  config = mkIf cfg.enable {
    # Prefer the config file over settings if both are set.
    services.corerad.configFile = mkDefault (writeTOML "corerad.toml" cfg.settings);

    systemd.services.corerad = {
      description = "CoreRAD IPv6 NDP RA daemon";
      after = [ "network.target" ];
      wantedBy = [ "multi-user.target" ];
      serviceConfig = {
        LimitNPROC = 512;
        LimitNOFILE = 1048576;
        CapabilityBoundingSet = "CAP_NET_ADMIN CAP_NET_RAW";
        AmbientCapabilities = "CAP_NET_ADMIN CAP_NET_RAW";
        NoNewPrivileges = true;
        DynamicUser = true;
        Type = "notify";
        NotifyAccess = "main";
        ExecStart = "${getBin cfg.package}/bin/corerad -c=${cfg.configFile}";
        Restart = "on-failure";
      };
    };
  };
}
{'fraction_non_alphanumeric': 0.1142801707411719, 'fraction_numerical': 0.009313154831199068, 'mean_word_length': 2.004079254079254, 'pattern_counts': {'":': 2, '<': 7, '<?xml version=': 0, '>': 7, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1878435', 'n_tokens_mistral': 1629, 'n_tokens_neox': 1520, 'n_words': 364}